
MIPS汇编语言（自学材料） 

CPU和指令集 

执行指令是CPU的主要工作 

不同的CPU有不同的指令集 

指令集架构Instruction Set Architecture (ISA). 

Intel 80x86 (Pentium 4), Intel IA64 

IBM/Motorola PowerPC (Macintosh), MIPS, ... 

精简指令集(RISC)的哲学 

 MIPS –— 最早一家生产出商用 RISC 架构的半导体公司 

MIPS 简单、优雅，不被细节所累 

MIPS 在嵌入式中广泛应用 

而作为CISC代表的x86较少应用到嵌入式市场, 更多地应用到PC上 

计算机系统的组成结构 

计算机系统由5大部分组成 
控制 

运算 

存储 

输入、输出 

寄存器是数据通路的一部分 

  

 Processor 

  

  

Computer 

 

 

Control 

Load (from) 

数据通路 

Program Counter (PC)

Instruction Register

Register File

ALU

Cache Memory

Data In

Address

4

Out

Rs

Rt
Rd

Control

Logic

0 $zero 

1 $at 

2 $v0 

3 $v1 

4 $a0 

5 $a1 

6 $a2 

7 $a3 

8 $t0 

9 $t1 

10 $t2 

11 $t3 

12 $t4 

13 $t5 

14 $t6 

15 $t7 

16 $s0 

17 $s1 

18 $s2 

19 $s3 

20 $s4 

21 $s5 

22 $s6 

23 $s7 

24 $t8 

Number Name Value 寄存器堆 

函数传递参数 

函数返回值 

保存寄存器 

临时变量寄存器 

Main Memory Program Counter (PC)

4

Address Instruction Register (IR)

0 13 0 4 0 0 Op-Code Rs Rt Rd Immediate

4 13 0 8 0 9 13 0 8 0 9

8 0 4 8 4 32

12 5 8 0 0 -1 Address name

16 5 8 0 0 -3 0 0 $zero

20 13 0 2 0 1 1 60 $at

24 0 0 0 0 12 2 8 $v0

28 0 0 0 0 0 3 0 $v1

32 3 4 5 4 0 4 17 $a0 ALU

36 6 3 7 1 2 5 0 $a1

40 0 0 0 0 0 6 8890 $a2

44 0 0 0 0 0 7 349 $a3 Rs

48 0 0 0 0 0 8 7 $t0 0

52 0 0 0 0 0 9 1 $t1

56 6 9 5 3 0 10 -1 $t2

60 0 0 0 0 456 11 0 $t3

64 0 0 0 0 2345 12 777 $t4

68 0 0 0 0 890 13 -777 $t5

72 0 0 0 0 -234 14 21 $t6

76 0 0 0 0 67 15 17 $t7

80 0 0 0 0 -111 16 -349 $s0

84 0 0 0 0 6612 17 66 $s1

88 0 0 0 0 4563 18 78 $s2

92 0 0 0 0 0 19 98 $s3

96 0 0 0 0 -1 20 34 $s4

100 0 0 0 0 -223 21 12 $s5

104 0 0 0 0 4567 22 56 $s6 Rt

108 0 0 0 0 -36 23 123456 $s7 7

112 0 0 0 0 1 24 89999 $t8 4

116 0 0 0 0 3 25 -345678 $t9

120 0 0 0 0 2 26 $k0

124 0 0 0 0 -36 26 $k1

128 0 0 0 0 23 28 $gp

132 0 0 0 0 -99 29 400000 $sp

厖 0 0 0 0 0 30 $fp

1,087,418,233 0 0 0 0 0 31 48 $ra

ALU-OUT

7

Reg. File

Control

Logic

 Reg. File[Rd] = Rs operation Rt 

寄存器传送的控制逻辑 

 
IR = Mem[PC] 

PC = PC + 4 

 
Decode Instruction 

Read from Reg. File 
lw   or  sw 

 
 

Memory[Address] = Rt 

sw 

lw 

R-Type 
If (Rs == 0 ) then 

PC = PC + Offset 

beqz 

 Address = Rs + Offset 

Reg. File[Rt] = Memory[Address] 

内存布局 

Text: 程序代码段 

Static data: 全局变量 
例如, C语言中的静态变量, 常数数组和串 

$gp 寄存器初始地址±偏移量寻址本段内存 

Dynamic data: 堆 
例如, C中的malloc, Java中的new 

Stack: 栈，自动存储区 

寻址模式回顾 

Lw    $t0, 0($2) 

Lw    $t1, 4($2) 

Sw   $t1, 0($2) 

Sw   $t0, 4($2) 

Higher-Level Language 
Program (e.g.  C) 

Assembly Language 
Program (e.g.  MIPS) 

Machine Language 
Program (MIPS) 

Hardware Architecture Description 
(e.g.  block diagrams)  

Compiler 

Assembler 

Machine 
Interpretation 

temp = v[k]; 

v[k] = v[k+1]; 

v[k+1] = temp; 

0000 1001 1100 0110 1010 1111 0101 1000 

1010 1111 0101 1000 0000 1001 1100 0110  

1100 0110 1010 1111 0101 1000 0000 1001  

0101 1000 0000 1001 1100 0110 1010 1111  

Logic Circuit Description 
(Circuit Schematic Diagrams) 

Architecture 
Implementation 

汇编语言在层次结构中的位置 

更紧凑、更高效的代码 

有时比编译优化更有效 

  
 12 

Label：  Op-Code Dest.,  S1,   S2  #Comments  

  move  $a0,   $0  # $a0 = 0 

  li  $t0,    99  # $t0 = 99 

loop:   

  add   $a0,   $a0, $t0 # $a0 = $a0 + $t0 

  addi  $t0,    $t0, -1 # $t0 = $t0 - 1 

  bnez  $t0,    loop # if ($t0  != zero) branch to loop  

  li  $v0,   1 # Print the value in $a0 

  syscall 

  li  $v0,   10 # Terminate Program Run 

  syscall 

 

MIPS 汇编语言程序示例 



  
 13 

MIPS指令集 

算术、逻辑和移位指令 

存/取指令 

条件分支指令 

函数调用指令 

MIPS指令字格式 

R- Format（ Register ） 

 

 

 

I- Format（ Immediate ） 

 

 

 

J- Format（Jump） 

 

Op-Code          Rs           Rt           16 - Bit  Immediate Value 

         6               5 5   16 

Op-Code          Rs           Rt           Rd        shamt           func 

         6             5              5 5 5                  6 

Op-Code                      26 Bit Current Segment Address 

         6                26 

伪指令 

取地址      la  $s0, table 

取立即数  li   $v0, 10 

移动                  move $t8, $sp 

乘   mul  $t2, $a0, $a1 

除   div   $s1, $v1, $t7 

求余  rem  $s2, $v1, $t7  

取反  neg  $s0, $s0 

MIPS 寄存器堆 

寄存器命名规范约定 

$0    : 常量0 

$v0  : 函数返回值 

$a0  : 函数传递参数 

$t0   : 临时变量寄存器 

$s0  : 保存寄存器 

$sp  : 栈指针 

$ra   : 返回地址 

MIPS汇编指令和指令字 

MIPS指令语法、变量和注释 

MIPS汇编中的算术、逻辑和移位运算 

MIPS汇编中的数据存取 

MIPS汇编中的分支和循环 

MIPS汇编中的函数 

MIPS汇编中的寄存器规范约定 

存储程序概念和MIPS指令字 

MIPS 指令 

指令语法 

操作符， 目标，源1，源2 

1) 操作名称（“操作符”）  

2) 操作结果 (“目标”) 

3) 1st 操作数 (“源1”) 

4) 2nd 操作数 (“源2”) 

语法是固定的 

1 个操作符, 3 个操作数 

通过约定好的统一的规则使硬件实现更简单 

汇编语言中, 每一条语句就是执行指令集中的一条简单指令 

与C语言 (及其他大多数高级程序语言)不同, 每行汇编代码

仅执行一条汇编指令 

C 或 Java 等高级语言执行的指令数与所做的操作(=, +, -, *, /)有关 

MIPS汇编变量: 寄存器（Registers） (1/2) 

与高级程序设计语言C或Java不同, 汇编语言不使用变量 
Why not? 保持硬件的简单 

汇编语言的操作数都是（寄存器） registers 
直接在硬件上实现的，数量有限 

所有的操作都只能在寄存器上完成! 

优势: 寄存器是直接用硬件实现的, 它们一定很快！ 

劣势: 寄存器是直接在硬件上实现的, 它们预先就规定好
了固定的数目！ 
解决: MIPS 代码必须仔细编写，以便能够合理、有效地使用寄存器 

MIPS有32个寄存器 
为什么32?  简单源自规整！ 

每一个MIPS寄存器的宽度为32bits 
MIPS中，由32个二进制位组合在一起称为一个 字  

MIPS汇编变量:寄存器（Registers） (2/2) 

从0到31给32个寄存器编号 

每个寄存器除了有编号外，还有自己的名字 

按照编号的引用方式: 

$0, $1, $2, … $30, $31 

约定：每个寄存器都取一个名字以便写代码时更方便 

$16 - $23  $s0 - $s7 

  (与C语言中的变量对应) 

$8 - $15，$24-$25  $t0 - $t7, $8-$9 

  (与临时变量对应) 

通常情况下, 使用名字来指定寄存器，这样可以增加代

码的可读性 

C, Java 变量 vs. 寄存器 

C 语言(以及其他大多数高级程序语言) 中所有

变量需要事先声明成为一个特定的类型 

例如:   
int fahr, celsius;  

char a, b, c, d, e; 

所有变量只能表达它被声明的那个类型的一个

值 (int与char用法不同，不能混淆) 

在汇编语言里, 寄存器没有数值类型; 只能通过

代码来决定这些寄存器中的数值应该如何使用

和处理 

注释 

另一个增加代码可读性的方法: 注释 

 # 被用来做MIPS的注释 

任何从# 后开始到行末的内容都被视为注释内容，并被忽略 

注意: 与C语言的区别 

C 还有一种注释格式 

/* comment */  

这种方式可以跨越多行加注释 

MIPS汇编指令和存储格式 

MIPS指令语法、变量和注释 

MIPS汇编中的算术、逻辑和移位运算 

MIPS汇编中的数据存取 

MIPS汇编中的分支和循环 

MIPS汇编中的函数 

MIPS汇编中的寄存器规范约定 

存储程序概念和MIPS指令字 

MIPS 整数加/减 

加法 
例子: add $s0,$s1,$s2 (in MIPS)，相当于: a = b + c (in C) 

减法 
例子：sub $s3,$s4,$s5 (in MIPS)，相当于：d = e - f (in C) 

C 语言中一条语句 a = b + c + d - e 

拆成多条汇编指令 

add $t0, $s1, $s2 # temp = b + c 

add $t0, $t0, $s3 # temp = temp + d 

sub $s0, $t0, $s4 # a = temp - e 

C中每一行语句可以拆成许多行的MIPS 
编译时，每行”#”后面的语句都会被当成注释忽略 

另一条C语言的语句：f = (g + h) - (i + j) 

使用临时变量寄存器 

add $t0,$s1,$s2 # temp = g + h 

add $t1,$s3,$s4 # temp = i + j 

sub $s0,$t0,$t1 # f=(g+h)-(i+j) 

  
 26 

寄存器 Zero和立即数 

一个特别的立即数, 数字0, 在代码中频繁出现 

定义寄存器 zero ($0 or $zero) 
add $s0,$s1,$zero (in MIPS) 

 f = g (in C) 

考虑指令 add $zero,$zero,$s0 

立即数是数值常量 
在代码中频繁出现，所以专门针对立即数定义了一些指令 

立即数加 
 addi $s0,$s1,10 (in MIPS) 

 f = g + 10 (in C) 

语法与add指令类似，除了最后一个参数用数值代替了寄存器 

MIPS中没有立即数的减法 
尽可能地保持指令集越小越好  

addi …, -X = subi …, X => 所以没有subi 

例子：addi $s0,$s1,-10 (in MIPS) 

     f = g - 10 (in C) 

算术运算中的溢出 

发生溢出是由于计算机有限的数值表示引起的 

例如（4位无符号数） 

  +15           1111 

    +3           0011 

  +18         10010 

有些语言会自动检测出异常(Ada),  有些不会(C) 

MIPS有2种(+,-)运算指令，每一种指令又有两种数值方式 
下面的 可以检测出溢出异常 
 add (add) 
 add immediate (addi) 
 subtract (sub) 

下面的 不会检测出溢出异常 
 add unsigned (addu) 
 add immediate unsigned (addiu) 
 subtract unsigned (subu) 

  编译器会自动挑选合适的运算指令类型 

MIPS中的C编译器会使用 
addu, addiu, subu 

但是如果我们只有4位寄存器，
存不了5位, 那么这个结果就是
0010, 是+2, 结果就错了 

 

不检查溢出异常 

位操作 

把寄存器中的值拆开来，看成是32个单独的位，

每个位表示一个2进制数值 

因为寄存器是由32个位组成的，有时候可能会想去单

独处理其中某个或某些位，而非整个数值 

逻辑和移位操作 

逻辑操作符 

两种基本的逻辑操作符 

AND: 当两个数都为1时输出1 

OR:   至少有一个为1时输出1 

逻辑指令的语法 
操作符   结果寄存器, 1st 操作数（寄存器）, 2nd 操作数（寄存器/立即数） 

操作符（指令名） 

and, or: 这两种指令的第三个参数（即2nd 操作数）都是寄存器 

andi, ori: 这两种指令的第三个参数（即2nd 操作数）都是立即数 

按位与、按位或 

C: 按位与& ，(比如, z = x & y;) 

C: 按位或 |，(比如, z = x | y;) 

移位指令  

将一个字的所有位向左或向右移动一定的位数 

移位指令语法 
     操作符    结果寄存器, 1st 操作数（寄存器）, 移位量（<32的常量/寄存器） 

MIPS移位指令 
sll (逻辑左移):左移并且补0，位移量为立即数 

（C中的<<） 

srl (逻辑右移):右移并且补0 ，位移量为立即数 

（C中的>>） 

sra (算术右移): 右移并且在空位做符号扩展填充，位移量为立即数 

 

 

MIPS使用变量值的移位指令 
sllv, srlv, srav 

与使用立即数位移量的处理方式类似, 移位量存储在寄存器中 

好的编译器会在乘以2n时，自动将乘法转换为移n位操作 
   a *= 8; (in C )  →  sll   $s0,$s0,3 (in MIPS) 

类似的, 算术右移n位就是除以2n 

MIPS汇编指令和指令字 

MIPS指令语法、变量和注释 

MIPS汇编中的算术、逻辑和移位运算 

MIPS汇编中的数据存取 

MIPS汇编中的分支和循环 

MIPS汇编中的函数 

MIPS汇编中的寄存器规范约定 

存储程序概念和MIPS指令字 

  
 32 

MIPS汇编操作数: 内存 

C变量被映射成寄存器; 如果是大型数据结构，例如数组

该怎么办呢? 

MIPS 算术指令只能操作寄存器, 不能直接操作内存 

数据存取指令在内存与寄存器之间传输数据 

内存到寄存器  

寄存器到内存 

  

 Processor 

  

  

Computer 

 

 

Control 

Datapath 

Registers 

Memory 

 

 

 

Devices 

 

 

Input 

Output Load (from) 

Store (to) 

寄存器被包含在处理器的

数据通路中;  如果操作数

在内存中, 我们必须首先

将它们转移到处理器上才

能进行操作, 然后当我们

处理完成之后，再将其送

回内存 

数据存取: 内存到寄存器 

要指定访问内存的具体地址, 需要两个数据值 
一个指向内存某地址的指针和一个数字偏移量 

我们需要的内存地址通常是这两个值相加所得的和 

例子: 8($t0) 

根据$t0中的值找到内存中的对应地址, 再加上 8 bytes 

Load 指令语法 
 操作码，寄存器，数值偏移量(寄存器)   

Load指令操作码：lw 

例子: lw $t0,12($s0) 

 先取出$s0中的指针的值, 加上12 bytes 的偏移量, 将这两个值相加即

得到我们要访问的内存地址，然后，将内存中该地址存储的数值送进
$t0。 

$s0 称为基址寄存器，12 称为偏移量 

偏移量经常用来访问数组或结构体中的元素: 基址寄存器指向该数组或
结构体的首地址 (注意偏移量必须是一个 在编译时就已经有确定值的
数值常量) 

数据存取: 寄存器到内存 

将寄存器中的数值写回内存中去 

Store与Load指令的语法格式是完全一样的 

操作码，寄存器，数值偏移量(寄存器) 

MIPS 指令名： sw 

例子: sw $t0,12($s0) 

 先取出$s0中存放的指针的值, 加上12 bytes, 计算

出两个值的和，即是要访问的内存地址，然后，将$t0

中的值送入内存中该地址存放 

记住: “Store 进内存” 

指针 vs. 值 

一个寄存器中可以存储32-bit的数值 

它可以是一个 (signed) int, 一个unsigned int, 一个

pointer (内存地址), 或是其它的类型 

如果你写了add $t2,$t1,$t0 那么 $t0 与 $t1最好是可

以相加的，结果是有意义的值 

如果你写了lw $t2,0($t0)那么 $t0 里面一定要是一个地址

值(pointer) 

寻址: 字节 vs. 字 

内存中的每一个字都有一个地址, 类似于一个数组的索引 

早期的计算机按字编址，就像C语言中的数组下标 

Memory[0], Memory[1], Memory[2],  … 
 
 

访问字(4 bytes/word)的同时也需要访问8-bit 字节(byte) 

今天计算机按字节编址，32-bit (4 bytes) 字地址按 4递增 

Memory[0], Memory[4], Memory[8], … 

如果要选择C语言中的A[5]  

 g = h + A[5]   ; g: $s1, h: $s2, $s3: A的基地址  

首先，将数据从内存送到寄存器: 

    lw $t0,20($s3)  # $t0 gets A[5] 

然后，将它与h相加，结果送进g 
  add $s1,$s2,$t0  # $s1 = h+A[5] 
 

被称作“字”地
址 

字对齐 

字节地址和字地址 

无论是 lw 还是 sw, 基址寄存器的值与偏移量的和始终应该是4的倍数 

MIPS 要求所有字的首地址必须是4的倍数 

Alignment(字对齐): 对象的起始地址一定要是字长的整数倍 

 

  0      1     2     3 

 

 

 

 
 

 

 

 

对齐的 

未 

对 

齐 

的 

0, 4, 8, or Chex 

地址的最后一个16进制数: 

1, 5, 9, or Dhex 

2, 6, A, or Ehex 

3, 7, B, or Fhex 

寄存器vs. 内存 

变量数比寄存器数多怎么办? 

编译器会将最经常使用的变量保留在寄存器中 

不常使用的放在内存中: spilling 

为什么不把所有的变量都放在寄存器里? 

小就是快: 

寄存器比内存要快 

MIPS 计算型指令每执行一条指令时，可以从两次读操作取数据，计算

结果，一次写操作到寄存器中 

MIPS 数据迁移型指令在每一条指令中，对寄存器只有一次读或写操作 

如果把C语言中的表达式*x = *y 翻译成 MIPS汇

编的指令，怎么样才是正确的呢？ 

(假定x, y 指针分别存储在 $s0，$s1中) 

A: add $s0,   $s1, zero 

B: add $s1,   $s0, zero 

C: lw  $s0, 0($s1) 

D: lw  $s1, 0($s0) 

E: lw  $t0, 0($s1) 

F: sw  $t0, 0($s0) 

G: lw  $s0, 0($t0) 

H: sw  $s1, 0($t0) 

0: A 

1: B  

2: C 

3: D 

4: EF 

5: EG 

6: FE 

7: FH 

8: HG 

9: GH 

 

例子1 

例子2 

我们需要完成以下指令: 

   int x = 5; 

   *p = *p + x + 10; 

MIPS (假定 $s0 中存放 p, $s1等于x) 

addi $s1,$0,5  # x = 5 

lw   $t0,0($s0)  # temp = *p 

add  $t0,$t0,$s1        # temp += x 

addi $t0,$t0,10  # temp += 10 

sw   $t0,0($s0)  # *p = temp 

字节的存/取 

除了需要在内存与寄存器之间按字传送 (lw, sw)外, 
MIPS 还有按字节传送的指令: 
读字节: lb 

写字节: sb 

与lw, sw格式相同 

例如：lb $s0, 3($s1) 

把内存中的某个地址(“3” + s1中的地址值)所存储的数值拷贝到s0的
低地址字节上. 

那么32位中的其余24位填充什么呢? 
lb: 使用位扩展(或称为符号扩展)填充剩余24位 

 

 

 

 

有些情况下我们不想使用位扩展(char类型) 

 MIPS 不使用位扩展的指令: lbu 

x  
读取的字节 

… 拷贝到前24位“符号扩展” 

 

这一位 

xxxx xxxx xxxx xxxx xxxx xxxx  zzzzzzz 

MIPS汇编指令和指令字 

MIPS指令语法、变量和注释 

MIPS汇编中的算术、逻辑和移位运算 

MIPS汇编中的数据存取 

MIPS汇编中的分支和循环 

MIPS汇编中的函数 

MIPS汇编中的寄存器规范约定 

存储程序概念和MIPS指令字 

目前为止，我们见过的指令都是操作数据的指令 … … 

我们做成了一个计算器 

为了实现一个真正的计算机, 我们需要代码能够做一些

判断、决定和转跳 … 

C 和 MIPS都提供 labels标签 用来支持“goto” 在代

码中跳来跳去. 

C: 使用goto是非常可怕的; MIPS: 必须有goto! 

C 判断: if 语句 

C中有两种if语句 

if (condition) clause 

if (condition) clause1 else clause2 

我们对上面第二种if语句改写如以下 

 if  (condition) goto L1; 

     clause2; 

        goto L2; 

L1: clause1; 

 L2: 

不如if-else看上去优雅, 但是效果一样 

MIPS 判断指令 

MIPS中的判断指令 

beq   register1, register2, L1 

beq =“Branch if (registers are) equal”  

相当于C当中:  
 if  (register1==register2) goto L1 

MIPS中还有一个与其互补的指令 

bne   register1, register2, L1 

bne = “Branch if (registers are) not equal”  

相当于C当中:  
 if  (register1!=register2) goto L1 

称为条件分支 

  
 46 

MIPS ‘Goto’ 指令 

除了条件分支语句之外, MIPS 还有无条件分支 

   j label 

跳转指令: 跳转到标签label所在的代码，不需要满足任何条件 

与C中的goto语句用法相同   

 goto label 

从技术的角度来说, 它与以下写法相同 

beq $0,$0,label 

从C的if语句到MIPS 

C语言中的条件判断 

 if (i == j) f=g+h;  

else f=g-h; 

假定使用如下的映射 
f: $s0 ；g: $s1；h: $s2；i: $s3；j: $s4 

Exit 

 i == j? 

f=g+h f=g-h 

(false)  
i != j 

(true)  
i == j 

MIPS代码 

    beq $s3,$s4,True    # branch i==j 

    sub $s0,$s1,$s2     # f=g-h(false) 

    j   Fin             # goto Fin 

True: add $s0,$s1,$s2   # f=g+h (true) 

Fin: 

注意:   编译器会在处理判断语句时自动为语句添加label去执行
分支，所以我们在C的if语句中找不到goto和label 

循环 

C中的循环; 假定A[] 是整型数组 

 do {       

     g = g + A[i];     

     i = i + j;     

 } while (i != h); 

假定使用以下映射 

  g, h, i, j, A[]的基址分别对应$s1,$s2,$s3,$s4,$s5 

MIPS 代码 

Loop:  sll $t1,$s3,2   #$t1= 4*i 

     add $t1,$t1,$s5 #$t1=addr A 

     lw  $t1,0($t1)  #$t1=A[i] 

     add $s1,$s1,$t1 #g=g+A[i] 

     add $s3,$s3,$s4 #i=i+j 

     bne $s3,$s2,Loop# goto Loop if i!=h 

C中有3种循环结构 
While；do… while；for 

每一种都可以用另外两种中的任一种等价表达, 上面的例子同样适用于
while 和 for 循环 

重写成 

 Loop: g = g + A[i]; 

  i = i + j; 

  if (i != h) goto Loop; 



  
 49 

MIPS中的不等式 

MIPS 不等式指令 
语法: slt reg1,reg2,reg3 

含义:“Set on Less Than”（‘set’指置为1） 

  if (reg2 < reg3) 

  reg1 = 1;  

 else reg1 = 0;  

如何表达这样的语句：if (g < h) goto Less;  

MIPS 代码（假定 g:$s0, h:$s1） 

 slt $t0,$s0,$s1 # $t0 = 1 if g<h  

bne $t0,$0,Less # goto Less 

                # if $t0!=0 

                # (if (g<h)) Less: 

寄存器 $0 的值总是 0, 所以 bne 和 beq 通常在slt指令后用寄存器0来做

分支判断 
   使用slt  bne 指令对表示if(… < …)goto… 

如何表达>, ≥? 增加新指令？ 

 

 

reg1 = (reg2 < reg3); 

不等式中的立即数和无符号数 

slt 指令的立即数版本: slti 

  if (g >= 1) goto Loop 

 Loop:  . . . 

slti $t0,$s0,1    # $t0 = 1 if 

                  # $s0<1 (g<1) 

beq  $t0,$0,Loop  # goto Loop 

                  # if $t0==0 

      # (if (g>=1)) 

 

无符号数不等式指令: sltu, sltiu 

    基于无符号数比较的结果置位 

C 

M 
I 
P 
S 

    使用slti  beq 指令对表示 if(… ≥ …)goto… 

MIPS 有符号 vs. 无符号 

比较以下指令执行后$t0和$t1的值  
($s0=FFFFFFFAH ，$s1=0000FFFAH) 

    slt  $t0, $s0, $s1 

    sltu $t1, $s0, $s1 

MIPS 中的有符号和无符号 

有符号扩展/无符号扩展 

(lb, lbu) 

没有(不得)溢出  

(addu, addiu, subu, multu, divu) 

有符号比较/无符号比较 

(slt, slti/sltu, sltiu) 

例子: C语言 Switch 语句 

根据k的取值从4个选项中选择1个，C代码如下：  
switch (k) { 

 case 0: f=i+j; break; /* k=0 */ 

 case 1: f=g+h; break; /* k=1 */ 

 case 2: f=g–h; break; /* k=2 */ 

 case 3: f=i–j; break; /* k=3 */ 

} 

先简化成if-else语句链 
if(k==0) f=i+j;  
 else if(k==1) f=g+h;  
   else if(k==2) f=g–h; 
     else if(k==3) f=i–j; 
变量映射：f:$s0, g:$s1, h:$s2, i:$s3, j:$s4, k:$s5 

MIPS 代码如下 
    bne  $s5,$0,L1   # branch k!=0 
    add  $s0,$s3,$s4 #k==0 so f=i+j 
    j    Exit        # end of case so Exit 
L1: addi $t0,$s5,-1  # $t0=k-1 
    bne  $t0,$0,L2   # branch k!=1 
    add  $s0,$s1,$s2 #k==1 so f=g+h 
    j    Exit        # end of case so Exit 
L2: addi $t0,$s5,-2  # $t0=k-2 
    bne  $t0,$0,L3   # branch k!=2 
    sub  $s0,$s1,$s2 #k==2 so f=g-h 
    j    Exit        # end of case so Exit 
L3: addi $t0,$s5,-3  # $t0=k-3 
    bne  $t0,$0,Exit # branch k!=3 
    sub  $s0,$s3,$s4 #k==3 so f=i-j  
Exit: 

第十六讲 

MIPS汇编指令和指令字 

MIPS指令语法、变量和注释 

MIPS汇编中的算术、逻辑和移位运算 

MIPS汇编中的数据存取 

MIPS汇编中的分支和循环 

MIPS汇编中的函数 

MIPS汇编中的寄存器规范约定 

存储程序概念和MIPS指令字 

C 函数 

main() { 

int i,j,k,m; 

... 

i = mult(j,k); ...  

m = mult(i,i); ... 

} 

/* really dumb mult function */ 

int mult (int mcand, int mlier){ 

int product; 

  product = 0; 

while (mlier > 0)  { 

 product = product + mcand; 

 mlier = mlier -1; } 

return product; 

} 

哪些信息是编译器和程序员

需要追踪和记录的？ 

什么指令可以实现这样的功能？ 

记录函数调用 

寄存器用来记录函数调用信息 

寄存器规范 

返回地址     $ra 

参数  $a0, $a1, $a2, $a3 

返回值         $v0, $v1 

局部变量     $s0, $s1, … , $s7 

还会用到栈！ 

支持函数功能的指令(1/3) 

 ... sum(a,b);... /* a,b:$s0,$s1 */ 

} 

int sum(int x, int y) { 

 return x+y; 

} 

 address 

1000  

1004  

1008  

1012  

1016  

 2000  

2004 

C 

M 
I 
P 
S 

MIPS中,所有指令都占4个字节,

并且像数据一样存在内存中（存
储程序概念），这里的地址是程
序存储的地址 

支持函数功能的指令(2/3) 

 ... sum(a,b);... /* a,b:$s0,$s1 */ 

} 

int sum(int x, int y) { 

 return x+y; 

} 

 address 

1000  

1004   

1008 

1012 

1016 

 2000 

2004 

C 

M 
I 
P 
S 

为什么用jr而不用 j? 

sum 可能会被多个函数调用，不能返回到某个固定的地址！
调用的函数一定会指定返回的地址 

add  $a0,$s0,$zero  # x = a 

add  $a1,$s1,$zero  # y = b  

addi $ra,$zero,1016 #$ra=1016 

j    sum     #jump to sum 

... 

sum: add $v0,$a0,$a1 

jr   $ra         # new instruction 

支持函数功能的指令(3/3) 

可同时执行转跳和存储返回地址的指令 
 jal， (jump and link ) 

   1008 addi $ra,$zero,1016  #$ra=1016 

1012 j sum       #goto sum 

可用单条指令代替 

          1008 jal sum  # $ra=1012,goto sum 

用jal使得函数调用更快，同时不需要了解代码读入内存的
地址细节 

jal 与 j 的语法相同 
  jal label 

jal 执行步骤其实应该是‘laj’(link and jump) 
第一步 (link):    将下一条指令地址存入$ra  (为什么下一条?) 

第二步 (jump): 向给定的label转跳 

寄存器转跳指令jr，可转跳至寄存器中存储的地址 

  jr register 

在函数调用中非常有用 

jal 指令将返回地址存储在寄存器($ra)中 

jr $ra 跳回该地址 

  
 60 

嵌套调用 

int sumSquare(int x, int y) { 
 return mult(x,x)+ y; 
} 

sumSquare被调用，而sumSquare又调用mult 
$ra中存储着 sumSquare 的返回地址，但是会在调用 mult时重写 
需要在调用mult 之前存储 sumSquare 返回地址 

需要在$ra之外存储相关信息！ 

当一个 C 程序运行时，有3块重要的内存区域被分配 
静态区（static）: 存储静态变量，一旦程序声明，直到程序执行结束 

                                 才会清除，比如C程序的全局变量 

堆（heap）:         动态声明的变量 

栈（stack）:        程序执行过程中使用的空间，可用来存储寄存器值 

Code  程序 

Static   程序声明的全局量 

Heap  比如, malloc(); C 指针等 

Stack  存储过程信息的空间 
$sp  

栈指针 

使用栈 

寄存器$sp始终指向栈空间最后被使用的位置 —— 栈指针 

使用栈的时候，对该指针减去需要的空间量，并向该空间填
写信息 

刚才的C例子 
 sumSquare:    

    addi $sp,$sp,-8    # space on stack 

    sw   $ra, 4($sp)  # save ret addr 

    sw   $a1, 0($sp)  # save y 

 

         add $a1,$a0,$zero  # prepare args: 2nd x  

       jal mult          # call mult 

 

         lw  $a1, 0($sp)  # restore y 

         add $v0,$v0,$a1    # mult()+y    

   lw $ra, 4($sp)  # get ret addr 

   addi $sp,$sp,8     # restore stack 

          jr $ra 

   mult: ... 

 

int sumSquare(int x, int y){ 
 return mult(x,x)+ y;} 

“push” 

“pop” 

  
 62 

函数调用规则 

函数调用的步骤 

将需要保存的值压入栈 

如果需要的话，指定参数 

jal 调用 

从栈中恢复相关的值 

调用过程中的规则 

通过 jal 指令调用, 使用一个  jr $ra 指令返回 

最多可接受4个入口参数 ， $a0, $a1, $a2, $a3 

返回值通常在 $v0 中 (如果需要，可以使用$v1) 

必须遵守寄存器使用规范  (即使是在那些只有你自己调用的函数中!)  

一个函数的基本结构 

 

entry_label:  
      addi $sp,$sp, -framesize 
      sw $ra, framesize-4($sp)  # save $ra 
      save other regs if need be     

    

 

 

      restore other regs if need be 
      lw $ra, framesize-4($sp)  # restore $ra 
      addi $sp,$sp, framesize  
      jr $ra 

      Body ...  (call other functions…) 

 

 ra 

memory 

MIPS汇编指令和指令字 

MIPS指令语法、变量和注释 

MIPS汇编中的算术、逻辑和移位运算 

MIPS汇编中的数据存取 

MIPS汇编中的分支和循环 

MIPS汇编中的函数 

MIPS汇编中的寄存器规范约定 

存储程序概念和MIPS指令字 

MIPS 寄存器分配 

    寄存器 0      $0  $zero 

汇编器预留               $1  $at 

返回值   $2-$3  $v0-$v1 

参数    $4-$7  $a0-$a3 

临时    $8-$15              $t0-$t7 

保存     $16-$23 $s0-$s7 

临时                $24-$25 $t8-$t9 

内核占用   $26-27              $k0-$k1 

全局指针   $28  $gp 

栈指针   $29  $sp 

帧指针   $30  $fp 

返回地址   $31  $ra 

$at:           编译器随时可能使用，最好不要用 

$k0-$k1:  操作系统随时会使用，最好不要用 

$gp, $fp:  可以不用理会 

寄存器规范 (1/2) 

CalleR: 发起调用的函数 

CalleE: 被调用的函数 

当callee需要返回时, caller 需要知道哪些寄存器的值可能已
经被更改了，还有哪些是保证不更改的 

寄存器规范: 一套被普遍遵从的规则 —— 在执行了一个函数
调用(jal) 后，哪些寄存器的值要保证不变，以及哪些可能

已经变了 

保存寄存器 

$0:  不能改变， 永远是0 

$s0-$s7:  如果被修改了需要恢复。如果 callee 由于各种原因

改变了这些值, 它必须在返回之前将这些寄存器的原始值恢复 

$sp: 如果被修改了需要恢复。 栈指针在jal 执行之前和之后

必须是指向的同一个地址，如若不然，caller 就无法从堆栈中
正常恢复数据了 

HINT – 所有保存寄存器都以 S开头！ 

寄存器规范 (2/2) 

易变寄存器 

$ra: 会改变。jal 会自动更改这个寄存器的值，Caller 需要将其
值保存在堆栈中 

$v0-$v1:会改变。始终保存最新的返回值 

$a0-$a3:会改变。这几个都是易变的参数寄存器，Caller 如果在

调用完Callee后还需要用到这些寄存器中的值，就需要在调用
Callee前，将这些值保存在自己的栈空间内 

$t0-$t9:会改变。 任何函数在任何时候都可以更新这些寄存器中

的值，Caller 如果在调用完Callee后还需要用到这些寄存器中的
值，就需要在调用Callee前将这些值保存在自己的栈空间内 

这样的规范和约定意味着什么？ 

如果函数 R 调用函数 E, 那么函数 R 必须赶在调用（jal）执行前，将所
有Callee可能会更新、而自己还会再用到的寄存器的值保存在自己的栈
空间内 

如果非得用到的话，函数 E 必须保存所有它将要更新的 S (保存) 寄存器
的值，并在函数返回（jr $ra）前及时恢复 

调用者和被调用者，都只需要保存它们使用的临时和保存寄存器值，
并非所有的寄存器都要保存 

MIPS汇编指令和指令字 

MIPS指令语法、变量和注释 

MIPS汇编中的算术、逻辑和移位运算 

MIPS汇编中的数据存取 

MIPS汇编中的分支和循环 

MIPS汇编中的函数 

MIPS汇编中的寄存器规范约定 

存储程序概念和MIPS指令字 

存储程序概念 

冯诺依曼计算机建立在 两个大原则之上 

指令与数值的表示形式一模一样 

全部程序可以被存储在内存中，像数据一样被读写 

简化计算机系统的软/硬件  

用于数据操作的内存技术完全适用于指令操作 

导致的结果1：编址 
所有的存储在内存中的东西都有一个地址，分支与跳转语句的
执行正是基于此（C中的指针存的正是内存中的地址值） 

对地址的随意使用会导致很难查找的bug 

有一个寄存器始终保存正在执行的指令地址: “Program 

Counter” (PC)，从根本上说就是一个指向内存的指针 

导致的结果2：二进制代码兼容性 

程序以二进制的形式给出，程序与特定的指令集绑定 

新机器想要运行旧程序 (“二进制代码”) 时，必须将程序按照
新的指令集进行编译 

导致 “向后兼容” 的指令集不断进化 

作为指令的数字 

现在我们处理的所有的数据都是按字来分配的(32位字长) 

每个寄存器是一个字 

lw , sw 每次只能访问内存中的一个字 

如何来表示指令呢? 

计算机只认识1和0, 所以“add $t0,$0,$0”对计算机来说没有意义 

MIPS 追求简单: 数据是按字存放的, 指令也按字存放吧！ 

一个字有32位, 我们把一个字分成几个“字段”(“fields”) 

每个“字段”用来提供指令的一部分信息 

可以定义不同的分配“字段”的方法, MIPS 基于简单原则, 定义

了以下3种指令格式的基本类型 
I-format(立即数格式) 

    当指令中有立即数的时候使用，包括lw、sw (偏移量是立即数) 以及分支

语句(beq and bne)。(但是这种格式不包含“移位”指令) 

J-format(跳转指令格式) 

  j ，jal 

R-format(寄存器格式) 

   适用于其他的指令 

R-Format 指令 

以位为单位定义各个“字段”的 大小 

  Opcode(6) rs(5) rt(5) rd(5) funct(6) shamt(5)  

每个字段都被看成是5-bit 或 6-bit 的无符号整数, 而不是一
个32-bit整数的一部分 

opcode: 与其他字段结合决定指令(等于0时代表所有R-Format指令) 

funct: 与opcode组合起来, 决定该条指令名(操作符) 

rs (Source Register): 通常指定存放第一个操作数的寄存器 

rt (Target Register): 通常指定存放第二个操作数的寄存器 

rd (Destination Register): 通常指定存放计算结果的寄存器 

shamt: 这个字段中存储执行移位运算时要移的位数(该字段在不进行移位

操作的指令中通常会置0) 

 注意3个寄存器字段: 

每个寄存器字段是5-bit, 可以用它来完整的表示出0-31之间的所有无符号整

数，这样每一个寄存器字段中的数值就是对应的32个寄存器中的一个 

当然有些特殊情况，我们会在后面提到 

R-Format 指令的例子 

MIPS 指令   
 
add   $8,$9,$10 
 
 

0 9 10 8 32 0  

opcode = 0  

funct = 32  

rd = 8 (目标结果)  

rs = 9 (第一 操作数) 

rt = 10 (第二 操作数) 

shamt = 0 (非移位指令) 

每个字段的
十进制表示 

每个字段的二进制表示 

000000 01001 01010 01000 100000 00000          

hex 

十六进制表示:         012A 402016 

十进制表示:            19,546,14410 

称为 机器语言指令 

I-Format 指令 

带立即数的指令? 
5-bit 的字段只能表示最大31的整数值: 立即数有可能大得多 
如果指令中需要立即数的话, 执行这条至少需要2个寄存器 

opcode(6) rs(5) rt(5)  immediate(16)  

只有一个字段与R-format不同, opcode 还在原来的位置不变 
opcode: 因为没有了funct字段, opcode 在I-format指令中可以唯一确定
一条指令 (R-format 用2个6-bit的字段而非一个12-bit字段来确定一条指令
的原因: 为了与其他指令格式保持一致) 

rs: 表示唯一的操作数寄存器 (如果有的话) 

rt: 存储计算结果的寄存器 (target register) 

立即数字段 

 addi, slti, sltiu, 立即数通过位扩展(符号扩展)的方式扩成32 位 

 16 bits  可以表示出216个不同的整数值 

 这么大的立即数在处理一些特别的指令(如lw 或 sw)时已经足够了, 
即使用slti指令，在大多数情况下也是没有问题的 

I-Format 指令 

I-Format 指令的例子 

MIPS 指令 
 

addi   $21,$22,-50 

 

opcode = 8  

rs = 22 (保存操作数的寄存器) 

rt = 21 (目标寄存器，存储结果值用) 

immediate = -50 (默认为十进制) 

8 22 21  -50  十进制表示 

二进制表示 

001000 10110 10101   1111111111001110  

十六进制表示:    0x22D5 FFCE16 

    十进制表示:      584,449,99810 

I-Format 指令的问题 

立即数太大怎么办？  
当需要的立即数在其字段内可以表示的时候，addi, lw, sw 和 slti 指

令执行时都没有问题 

但是如果太大，字段无法表示怎么办?在使用任何一个I-Format指令时，
我们都必须考虑：如果立即数是一个32-bit的数值该怎么办？ 

解决方案 
使用软件技巧 + 新的指令 

不改变现有指令: 只要加入一条新指令来帮忙 

新指令：lui   register, immediate 
Load Upper Immediate 

将一个16-bit的立即数存入寄存器的高16位，将寄存器的低16位全部置0 

 

 

 

立即数太大，
这条指令根本

放不进去 这样就没
问题啦 

例子:  addi   $t0,$t0, 0xABABCDCD 
改为:   lui    $at, 0xABAB 
     ori    $at, $at, 0xCDCD 
     add    $t0,$t0,$at 

每条I-format指令只有16-bit用来存放立即数 

 哪一条指令的机器码与十进制数35长得一样? 

1. add $0, $0, $0 

2. subu $s0,$s0,$s0 

3. lw $0, 0($0) 

4. addi $0, $0, 35 

5.  subu $0, $0, $0 

 

注意：只是长得一样，但是指令不是数字！！ 

   35 0 0   0    

0 32 0 0 0 0  

   8 0 0   35    

16 35 0 0 16 16  

0 35 0 0 0 0   

使用I-Format的分支语句: 程序计数器相对寻址 

opcode rs rt   immediate  

    opcode 指明指令是 beq 或 bne 

rs 和 rt 指明要比较的两个寄存器 

立即数字段? 

立即数只有16 bits，PC (程序计数器) 有32-bit 的指向内存的指针；立
即数无法表示出完整的内存地址 

通常使用的if-else, while, for等分支语句，一般循环体都较小 

函数调用与无条件跳转指令都会用到跳转指令(j and jal), 不是分支指令 

结论:   多数情况下，分支语句转跳时，PC 的变化值都相差不大 

在32-bit 指令格式中执行分支语句的解决方案: PC-相对寻址 

将16-bit立即数使用补码表示，在需要分支的时候与PC相加 

可以分支到PC± 215 字节的地方 

还能不能更好？ 

注意: 指令的起始地址都一定要是4的倍数(字对齐) 

和PC相加的立即数也应该是4的倍数！其实可分支到PC± 215 指令字 了 
(或者说 PC± 217 字节) 

PC = (PC + 4) + (immediate * 4) 

分支的例子 

MIPS 代码 
   

beq 分支指令是I-Format格式的 
opcode = 4 

rs = 9 (第一操作数) 

rt = 0 (第二操作数) 

immediate = ??? 

立即数字段: 

要和PC相加(或相减)的指令数, 是从分
支语句的下一条指令算起的 

在这条 beq 的分支中, immediate = 3 

4 9 0   3  

分支指令的十进制表示 

000100 01001 00000   0000000000000011  

分支指令的二进制表示 

 
1 

2 

3 

关于PC-相对寻址的问题 

如果代码移动了，分支字段的值会变么? 

如果目标地址与分支指令相差 > 215怎么办? 

我们为什么需要这么多寻址方式? 一个不行么? 

J-Format 指令 

在分支语句中, 假定不会分支到太远的地方, 所以可以指明PC的变化值 

对于一般的跳转指令 (j 和 jal), 是有可能跳到内存中任意一个地方的 

理想情况下, 可以直接给出一个32-bit的内存地址，告诉要跳到哪里 

J-Format 指令 opcode(6) target address(26)  

 保持 opcode 字段与 R-format 及 I-format 一样，维护一致性原则 

 把其他所有字段都加到一起，使能表示的地址尽量大 

 利用字对齐，可以表示出32-bit地址的28 bits 

 剩下的最高4位根据定义, 直接从PC取 

New PC = { PC[31..28], target address, 00 } 

 如果确实需要一个32-bit 地址, 就把它放进寄存器，使用jr指令 

第五讲：指令系统与MIPS汇编 

一．指令格式 

1. 指令系统概述 

2. 指令格式 

3. 寻址方式 

二．典型指令系统介绍 

1. 8086/8088指令系统 

2. MIPS指令系统 

3. CISC与RISC 

三．MIPS汇编语言 

1. 概述 

2. MIPS汇编指令和指令字 

3. MIPS汇编程序       

  
 82 

MIPS汇编程序 

 MIPS汇编语言语句 

 MIPS汇编语言程序模板 

 数据定义 

 内存对齐和字节序 

 系统调用 

 过程 

 参数传递和运行时栈 

汇编语言语句 

 MIPS汇编中的3类语句 

 通常一个语句一行 

1. 可执行指令 

 为处理器生成在运行时执行的机器码 

 指令告诉处理器该做什么 

2. 伪指令和宏 

 由汇编程序翻译成真正的指令 

 简化编程人员的工作  

3. 汇编伪指令 

 当翻译代码时为汇编程序提供信息 

 用来定义段、分配内存变量等 

 不可执行: 汇编伪指令不是指令集的一部分 

指令 

汇编语言指令格式 

 [标签:]   操作符   [操作数]    [#注释] 

标签: (可选) 

标记内存地址, 必须跟冒号 

通常在数据和代码段出现  

操作符 

定义操作 (比如 add, sub, 等) 

操作数 

指明操作需要的数据 

操作数可以是寄存器，内存变量或常数 

大多数指令有3个操作数 

 L1: addiu $t0, $t0, 1 #increment $t0 

注释 

注释是非常重要的! 

解释程序语句的目的 

程序语句的编写、修改时间和人 

解释程序中的数据，输入和输出 

解释指令序列和算法 

每个过程的开始需要注释 

 指出输入参数和过程的结果 

 描述过程的功能 

单行注释 

由‘ #’开头在1行内结束 

 MIPS汇编语言语句 

 MIPS汇编语言程序模板 

 数据定义 

 内存对齐和字节序 

 系统调用 

 过程 

 参数传递和运行时栈 

 

MIPS汇编程序 

程序模板 

# Title: Filename: 

# Author: Date: 

# Description: 

# Input: 

# Output: 

################# Data segment ##################### 

.data 

 . . . 

################# Code segment ##################### 

.text 

.globl main 

main: # main program entry 

 . . . 

li $v0, 10 # Exit program 

syscall 

.DATA, .TEXT, 和 .GLOBL 伪指令 

.DATA 伪指令 

定义程序的数据段 

程序的变量需要在该伪指令下定义 

汇编程序会分配和初始化变量的存储空间 

.TEXT 伪指令 

定义程序的代码段 

.GLOBL 伪指令 

声明一个符号为全局的 

全局符号可以被其它的文件引用 

用该伪指令声明一个程序的 main 过程 

程序在内存中的存放 

堆栈段 
0x7FFFFFFF 

动态区域 

静态区域 

代码段 

预留 

 

0x04000000 

0x10000000 

0 

 数据段 

16进制内存

地址 

堆栈伸缩 

 MIPS汇编语言语句 

 MIPS汇编语言程序模板 

 数据定义 

 内存对齐和字节序 

 系统调用 

 过程 

 参数传递和运行时栈 

MIPS汇编程序 

数据定义 

为变量的存储划分内存 

可能会有选择的为数据分配名字（标签） 

语法: 

 [名字:]    伪指令     初始值 [, 初始值]  . . . 

 

var1: .WORD    10 

所有的初始值在内存中以二进制数据存储 

数据伪指令 

.BYTE 伪指令 

以8位字节存储数值表 

 .HALF 伪指令 

以16位（半字长）存储数值表  

 .WORD 伪指令 

以32位（一个字长）存储数值表 

 .WORD w:n 伪指令 

将32位数值 w 存入 n 个边界对齐的连续的字中 

 .FLOAT 伪指令 

以单精度浮点数存储数值表 

 .DOUBLE 伪指令 

以双精度浮点数存储数值表 

字符串伪指令 

.ASCII 伪指令 

为一个ASCII字符串分配字节序列 

.ASCIIZ 伪指令 

与.ASCII 伪指令类似, 但是在字符串末尾增加 NULL字符 

字符串以NULL结尾，类似C语言 

.SPACE n 伪指令 

为数据段中 n 个未初始化的字节分配空间 

字符串中的特殊字符（按照 C 语言的约定） 

新行: \n            Tab:\t  引用: \” 

数据定义的例子 

.DATA 

var1:  .BYTE     'A', 'E', 127, -1, '\n' 

var2:  .HALF     -10, 0xffff 

var3:  .WORD     0x12345678 

Var4:  .WORD     0:10 

var5:  .FLOAT    12.3, -0.1 

var6:  .DOUBLE   1.5e-10 

str1:  .ASCII    "A String\n" 

str2:  .ASCIIZ   "NULL Terminated String" 

array: .SPACE    100 

如果初始值超过了值域上界，
汇编程序会报错 

MIPS汇编语言语句 

MIPS汇编语言程序模板 

数据定义 

内存对齐和字节序 

系统调用 

过程 

参数传递和运行时栈 

 

MIPS汇编程序 

内存可以被看成是带地址的字节数组 

字节编址: 地址指向内存中的一个字节 

字占据内存中4个连续的字节 

MIPS 指令和整数占据4个字节 

对齐: 地址是空间大小的整数倍 

字的地址是4的整数倍 

 地址的2位最低有效位必须是 00 

半字的地址是 2 的整数倍 

.ALIGN n 伪指令 

对下一个定义的数据做 2n 字节对齐 

内存对齐 

 

0 

4 

8 

12 

a
d

d
re

s
s

 

未对齐 

. . . 

对齐的字 

 

 

 

 
未对齐 

内存 

汇编程序为标签（变量）构建符号表 

为数据段的每一个标签计算地址 

例子  

 .DATA 

 var1:  .BYTE   1, 2,'Z' 

 str1:  .ASCIIZ "My String\n" 

 var2:  .WORD   0x12345678 

 .ALIGN  3 

 var3:  .HALF   1000 

符号表 

标签 

var1 

str1 

var2 

var3 

地址 

0x10010000 

0x10010003 

0x10010010 

0x10010018 

 

0 0 0 0 0 0 

var1 

1 2 'Z' 0x10010000 

str1 

'M' 'y' '  ' 'S' 't' 'r' 'i' 'n' 'g' '\n' 0 

0x12345678 0x10010010 

var2 (aligned) 

1000 

var3 (address is multiple of 8) 

0 0 Unused  

0 0 0 0 

Unused 

符号表 

处理器对一个字内的字节排序有两种方法 

小端字节排序 

内存地址 = 最低有效字节的地址 

例子: Intel IA-32, Alpha 

 

 

大端字节排序 

内存地址 = 最高有效字节的地址 

例子: SPARC, PA-RISC 

 

 

MIPS 可以操作以上两种字节序  

字节序和端 

Byte 0 Byte 1 Byte 2 Byte 3 

32-bit Register 

MSB LSB 

. . . . . . Byte 0 Byte 1 Byte 2 Byte 3 

a a+3 a+2 a+1 

Memory 

address 

 

Byte 3 Byte 0 Byte 1 Byte 2 Byte 3 

32-bit Register 

MSB LSB 

. . . . . . Byte 0 Byte 1 Byte 2 

a a+3 a+2 a+1 

Memory 

address 

MIPS汇编语言语句 

MIPS汇编语言程序模板 

数据定义 

内存对齐和字节序 

系统调用 

过程 

参数传递和运行时栈 

 

MIPS汇编程序 

  
 100 

系统调用 

程序通过系统调用实现输入/输出 

MIPS 提供一条特殊的 syscall 指令 

从操作系统获取服务 

使用 syscall 系统服务 

从 $v0寄存器中读取服务数 

从 $a0, $a1, 等寄存器中读取参数值（如果有） 

发送 syscall 指令 

从结果寄存器中取回返回值（如果有） 

Syscall 服务 

Service $v0 Arguments / Result 

Print Integer 1 $a0 = integer value to print 

Print Float 2 $f12 =  float value to print 

Print Double 3 $f12 = double value to print 

Print String 4 $a0 = address of null-terminated string 

Read Integer 5 $v0 = integer read 

Read Float 6 $f0 = float read 

Read Double 7 $f0 = double read 

Read String 8 $a0 = address of input buffer 

$a1 = maximum number of characters to read 

Exit Program 10 

Print Char 11 $a0 = character to print 

Read Char 12 $a0 = character read 

读和打印一个整数 

################# Code segment ##################### 

.text 

.globl main 

main:  # main program entry 

 li $v0, 5 # Read integer 

 syscall # $v0 = value read 

 

 move $a0, $v0 # $a0 = value to print 

 li $v0, 1 # Print integer 

 syscall 

 

 li $v0, 10 # Exit program 

 syscall 

读和打印一个串 

################# Data segment ##################### 

.data 

 str: .space  10 # array of 10 bytes 

################# Code segment ##################### 

.text 

.globl main 

main:  # main program entry 

 la $a0, str # $a0 = address of str 

 li $a1, 10 # $a1 = max string length 

 li $v0, 8 # read string 

 syscall 

 li $v0, 4 # Print string str  

 syscall 

 li $v0, 10 # Exit program 

 syscall 

例程 :  3整数求和 

# Sum of three integers 

# 

# Objective: Computes the sum of three integers.  

#     Input: Requests three numbers. 

#    Output: Outputs the sum. 

################### Data segment ################### 

.data 

prompt: .asciiz     "Please enter three numbers: \n" 

sum_msg: .asciiz     "The sum is: " 

################### Code segment ################### 

.text 

.globl main 

main: 

      la    $a0,prompt # display prompt string 

      li    $v0,4 

      syscall 

      li    $v0,5 # read 1st integer into $t0 

      syscall 

      move  $t0,$v0 

例程:  3整数求和（2） 

      li    $v0,5 # read 2nd integer into $t1 

      syscall 

      move  $t1,$v0 

      li    $v0,5 # read 3rd integer into $t2 

      syscall 

      move  $t2,$v0 

      addu  $t0,$t0,$t1 # accumulate the sum  

      addu  $t0,$t0,$t2 

      la    $a0,sum_msg # write sum message 

      li    $v0,4 

      syscall 

      move  $a0,$t0 # output sum 

      li    $v0,1 

      syscall 

      li    $v0,10 # exit 

      syscall 

例程 : 大小写转换 

# Objective: Convert lowercase letters to uppercase 

#     Input: Requests a character string from the user. 

#    Output: Prints the input string in uppercase. 

################### Data segment ##################### 

.data 

name_prompt: .asciiz "Please type your name: " 

out_msg: .asciiz "Your name in capitals is: " 

in_name: .space 31 # space for input string 

################### Code segment ##################### 

.text 

.globl main 

main: 

      la    $a0,name_prompt # print prompt string 

      li    $v0,4 

      syscall 

     la    $a0,in_name # read the input string 

      li    $a1,31 # at most 30 chars + 1 null char 

      li    $v0,8 

      syscall 

例程 : 大小写转换（2） 

      la    $a0,out_msg      # write output message 

      li    $v0,4 

      syscall 

      la    $t0,in_name 

loop: 

      lb    $t1,($t0) 

      beqz  $t1,exit_loop    # if NULL, we are done 

      blt   $t1,'a',no_change 

      bgt   $t1,'z',no_change 

      addiu $t1,$t1,-32      # convert to uppercase: 'A'-'a'=-32        

    sb    $t1,($t0) 

no_change: 

      addiu $t0,$t0,1        # increment pointer  

      j     loop 

exit_loop: 

      la    $a0,in_name      # output converted string 

      li    $v0,4 

      syscall 

      li    $v0,10           # exit 

      syscall 

MIPS汇编语言语句 

MIPS汇编语言程序模板 

数据定义 

内存对齐和字节序 

系统调用 

过程 

参数传递和运行时栈 

MIPS汇编程序 

参数: 

$a0 = v[]的地址  

$a1 = k,   

返回地址在 $ra 中 

观察 swap 过程 (C程序) 

翻译成 MIPS 汇编语言 

void swap(int v[], int k) 

{  int temp; 

 temp = v[k] 

 v[k] = v[k+1]; 

 v[k+1] = temp; 

} 

swap: 

 sll $t0,$a1,2   # $t0=k*4 

 add $t0,$t0,$a0 # $t0=v+k*4 

 lw  $t1,0($t0)  # $t1=v[k] 

 lw  $t2,4($t0)  # $t2=v[k+1] 

 sw  $t2,0($t0)  # v[k]=$t2 

 sw  $t1,4($t0)  # v[k+1]=$t1 

 jr  $ra # return 

过程 

调用 / 返回 序列 

调用swap过程: swap(a,10) 

将数组a 的地址和10作为参数传递 

调用swap过程，保存返回地址 $31 = $ra 

执行swap过程 

返回对返回地址的控制 

swap: 

 sll $t0,$a1,2 

 add $t0,$t0,$a0 

 lw  $t1,0($t0) 

 lw  $t2,4($t0) 

 sw  $t2,0($t0) 

 sw  $t1,4($t0) 

 jr  $ra 

 la   $a0, a 

 li   $a1, 10 

 jal  swap 

# return here 

 . . . 

调用 

addr a $a0=$4 

10 $a1=$5 

ret addr $ra=$31 

. . . 

. . . 

寄存器 

寄存器 $31 

是返回地址寄存器 

JAL 和 JR的细节 

地址 指令                          汇编语言 

 

00400020 lui $1, 0x1001  la   $a0, a 

00400024 ori $4, $1, 0 

00400028 ori $5, $0, 10 li   $a1,10 

0040002C jal 0x10000f jal  swap 

00400030  . . . # return here 

 

 

  swap: 

0040003C sll $8, $5, 2 sll $t0,$a1,2 

00400040  add $8, $8, $4 add $t0,$t0,$a0 

00400044  lw  $9, 0($8) lw  $t1,0($t0) 

00400048  lw  $10,4($8) lw  $t2,4($t0) 

0040004C  sw  $10,0($8) sw  $t2,0($t0) 

00400050  sw  $9, 4($8) sw  $t1,4($t0) 

00400054  jr  $31 jr  $ra 

伪直接寻址 

PC = imm26<<2 

0x10000f << 2 

= 0x0040003C  

 

0x00400030 $31 

过程的指令 

Instruction Meaning Format 

jal label $31=PC+4, jump op6 = 3 imm26 

jr Rs PC = Rs op6 = 0 rs5 0 0 0 8 

jalr Rd, Rs Rd=PC+4, PC=Rs op6 = 0 rs5 0 rd5 0 9 

 JAL (Jump-and-Link) ：调用指令 

 在 $ra = PC+4 中保存返回地址并跳转到相应的过程 

 寄存器 $ra = $31 被 JAL 用来保存返回地址 

 JR (Jump Register) ：返回指令 

 跳转到在 寄存器Rs (PC = Rs)中存储的地址所在指令 

 JALR (Jump-and-Link Register) 

 在Rd = PC+4中存储返回地址, 

 跳转到在寄存器Rs (PC = Rs)中存储的地址所在过程 

 用于调用方法(地址仅在运行时可知) 

 MIPS汇编语言语句 

 MIPS汇编语言程序模板 

 数据定义 

 内存对齐和字节序 

 系统调用 

 过程 

 参数传递和运行时栈 

 

MIPS汇编程序 

参数传递 

汇编语言中的参数传递比高级语言中复杂 

汇编语言中 
将所有需要的参数放置在一个可访问的存储区域 

然后调用过程  

会用到两种类型的存储区域 
寄存器: 使用通用寄存器 (寄存器方法) 

内存: 使用栈 (栈方法) 

参数传递的两种常用机制 
值传递: 传递参数值 

引用传递: 传递参数的地址 

参数传递（续） 

按照约定, 参数传递通过寄存器实现 
$a0 = $4 .. $a3 = $7 用来做参数传递 

$v0 = $2 .. $v1 = $3 用来表示结果数据 

其它的参数/结果可以放在栈中 

运行时栈用于 
不适合使用寄存器时用来存储变量/数据结构 

过程调用中保存和恢复寄存器 

实现递归 

运行时栈通过软件规范实现 
栈指针 $sp = $29 (指向栈顶) 

帧指针 $fp = $30 (指向过程帧) 

栈帧 

栈帧是栈的一段，包含 … 
保存的参数，寄存器和本地数据结构 

也称为活动帧或活动记录 

帧通过调整指针压入和弹出 
栈指针 $sp = $29 和帧指针 $fp = $30 

减指针 $sp 分配堆栈帧, 加指针释放 

Frame f() 

Stack 

↓ 
 

stack grows 
downwards 

$fp 

$sp 

Frame f() 

Stack 

allocate 
stack frame 

Frame g() 
$fp 

$sp 

f 
c
a
ll

s
 g

 

g
 r

e
tu

rn
s
 Frame f() 

Stack 

↑ 
 

free stack 
frame 

$fp 

$sp 

$fp 

arguments 

saved $ra 

saved 
registers 

local data 
structures 

or variables 
$sp 

保存寄存器 

过程调用时需要保存寄存器 

栈可用来保存寄存器值 

什么寄存器需要保存？ 

在被调过程修改，而调用过程还会再用到的寄存器 

谁来保存寄存器值？ 

被调过程: 模块化代码的首选方法 

 寄存器保存在被调过程内部实现 

通常约定, 寄存器 $s0, $s1, …, $s7 需要被保存 

同样, 寄存器 $sp, $fp, 和 $ra 需要被保存 

选择排序 

例子 

选择排序过程 

# Objective: Sort array using selection sort algorithm 

#     Input: $a0 = pointer to first, $a1 = pointer to last 

#    Output: array is sorted in place 

########################################################## 

sort: addiu $sp, $sp, -4 # allocate one word on stack 

 sw $ra, 0($sp) # save return address on stack 

top: jal max # call max procedure 

 lw $t0, 0($a1) # $t0 = last value 

 sw $t0, 0($v0) # swap last and max values 

 sw $v1, 0($a1) 

 addiu $a1, $a1, -4 # decrement pointer to last 

 bne $a0, $a1, top # more elements to sort 

 lw $ra, 0($sp) # pop return address 

 addiu $sp, $sp, 4 

 jr $ra # return to caller 

找最大值过程 

# Objective: Find the address and value of maximum element 

#     Input: $a0 = pointer to first, $a1 = pointer to last 

#    Output: $v0 = pointer to max,   $v1 = value of max 

########################################################## 

max: move $v0, $a0 # max pointer = first pointer 

 lw $v1, 0($v0) # $v1 = first value 

 beq $a0, $a1, ret # if (first == last) return 

 move $t0, $a0 # $t0 = array pointer 

loop: addi $t0, $t0, 4 # point to next array element 

 lw $t1, 0($t0) # $t1 = value of A[i] 

 ble $t1, $v1, skip # if (A[i] <= max) then skip 

 move $v0, $t0 # found new maximum 

 move $v1, $t1 

skip: bne $t0, $a1, loop # loop back if more elements 

ret: jr $ra  

递归过程示例 

int fact(int n) { if (n<2) return 1; else return (n*fact(n-1)); } 

fact: slti $t0,$a0,2 # (n<2)? 

 beq $t0,$0,else # if false branch to else 

 li $v0,1 # $v0 = 1 

 jr $ra # return to caller 

else: addiu $sp,$sp,-8 # allocate 2 words on stack 

 sw $a0,4($sp) # save argument n 

 sw $ra,0($sp) # save return address 

 addiu $a0,$a0,-1 # argument = n-1 

 jal fact # call fact(n-1) 

 lw $a0,4($sp) # restore argument 

 lw $ra,0($sp) # restore return address 

 mul $v0,$a0,$v0 # $v0 = n*fact(n-1) 

 addi $sp,$sp,8 # free stack frame 

 jr $ra # return to caller 


